CROATICA CHEMICA ACTA CCACAA 46 (3) 199—207 (1974)

YU ISSN 0011—1643
CCA-836
518.1:539.19
Note

Computer Program for the Evaluation of Overlap Integrals

*J. Zupan**

Chemical Institute »Boris Kidrič«, 61000 Ljubljana, Slovenia, Yugoslavia

Received May 24, 1974

The program written in FORTRAN IV for evaluating the overlap integrals of σ, π, δ, or φ type between two s, p, d, or f Slater AO's with quantum number 1 to 6 is described and the computational method is discussed.

INTRODUCTION

Several papers on the evaluation of overlap integrals[1-5] (OI) have appeared after Mulliken and coworkers[6] published the master formulas and tables for various overlap integrals in 1949. Together with the development of fast computers many very sophisticated programs and routines for different kinds of MO calculations, including the evaluations of OI's as well, have been written[7]. Unfortunately, such programs are dealing mainly with the first and second row elements and therefore do not allow calculations of the OI's of higher orders, or the OI's are not easily picked out from such a complex program.

In the present work a FORTRAN IV routine for evaluating σ, π, δ, or φ type OI's between two Slater AO's is described. The maximum quantum number for s, p, d, or f AO's is 6, therefore 348** different types of OI's can be evaluated using this routine.

METHOD OF CALCULATION

The OI for two Slater type AO's can be written as follows[6]:

$$S(n_a, AO_a, n_b, AO_b; p, t) = N_{ab} \int_0^\infty \int_{-1}^1 (\xi + \eta)^{s_a} (\xi - \eta)^{s_b}$$

$$P(\xi, \eta)\, e^{-p(\xi + \eta t)}\, d\xi\, d\eta \qquad (1)$$

where $P(\xi, \eta)$ is a polynomial

$$P(\xi, \eta) = \sum_i^L \alpha_i \xi^{l_i} \eta^{k_i}$$

and the integral over ξ and η can be expressed in terms of integrals A and B;

*Part of this work was done at the Institute »Jožef Stefan«, University of Ljubljana, Ljubljana.

** 36 ssσ, 30 spσ, 24 sdσ, 18 sfσ, 25 ppσ, 20 pdσ, 15 pfσ, 16 ddσ, 12 dfσ, 9 ffσ, 25 ppπ, 20 pdπ, 15 pfπ, 16 ddπ, 12 dfπ, 9 ffπ, 16 ddδ, 12 dfδ, 9 ffδ, 9 ffφ.

$$\int_0^\infty \int_{-1}^{1} P(\xi, \eta)\, e^{-p(\xi + \eta t)}\, d\xi\, d\eta = \sum_i^L \alpha_i A(l_i, p)\, B(k_i, pt)$$

$P(\xi, \eta)$ depends on the type of overlap ($\sigma, \pi, \ldots$), the quantum numbers ($n = 1, \ldots, 6$), and AO's (s, p, ...). The polynomial $P(\xi, \eta)$ can easily be obtained from Lofthus formulas by derivation[5]. The exponents s_a and s_b are mantissas of noninteger quantum numbers, *i. e.* they are equal to 0.7 and 0.2 for n equal to 4 (3.7) and 6 (4.2), respectively, and are zero for all other cases ($n = 1, 2, 3, 5$). Further, the OI (1) can be expanded[5]:

$$S = N_{ab} \int_0^\infty \int_{-1}^{1} \left(1 + \frac{\eta}{\xi}\right)^{s_a} \left(1 - \frac{\eta}{\xi}\right)^{s_b} .$$

$$P(\xi^{1 + s_a + s_b}, \eta)\, e^{-p(\xi + \eta t)}\, d\xi\, d\eta$$

$$= N_{ab} \int_0^\infty \int_{-1}^{1} \sum_{m=0}^{\infty} C_m \left(\frac{\eta}{\xi}\right)^m P(\xi^{1 + s_a + s_b}, \eta)\, e^{-p(\xi + \eta t)}\, d\xi\, d\eta$$

$$= N_{ab} \sum_{m=0}^{\infty} C_m \sum_{i=0}^{L} \alpha_i A(l_i + s_a + s_b - m, p)\, B(k_i + m, pt) \tag{2}$$

In the present form (2) the OI's are computed by the described routine. The integrals $A(\alpha, p)$ and $B(\beta, pt)$ are defined in the literature[1-6]. The calculation of the integral $A(\alpha, p)$ for noninteger values of the argument α is described in the Appendix I. For the calculation of the integral $B(\beta, pt)$ see for example[8]. The expansion over m is necessary in the cases of noninteger quantum numbers when s_a or s_b are different from zero. In all other cases the equation (2) is modified to:

$$S = N_{ab} \sum_{i=0}^{L} \alpha_i A(l_i, p)\, B(k_i, pt) \tag{2a}$$

DESCRIPTION OF THE PROGRAM

The subroutine ØVERLAP is completely self-contained (composed of three subroutines ØVERLAP, DER and DUMP, and of two functions A and B) and communication to it is solely through the argument list. The entrance to the subroutine can be achieved by:

CALL ØVERLAP (N1, N2, LØR1, LØR2, P, T, FACTØR, SØVE, IS, IØ). The

The meaning of the parameters is described in the comments at the beginning of the subroutine ØVERLAP. The Lofthus formulas[5] from which the OI can be obtained are stored in DATA statements in the twodimensional array ØV (IS, J) as follows:

— each formula in the separate row, indexed with IS (*i. e.* IS = 1 for s-sσ, IS = 2 for s-pσ and so on)

— the first number in each row, ØV (IS, 1), contains the normalisation constant of the Lofthus integral $s(n_1 l_1 m n_2 l_2 m)$[5]

— in the second place, ØV (IS, 2) = 3.N + 2, is stored, where N is the number of products of A and B integrals in each Lofthus integral $s(n_1 l_1 m,\ n_2 l_2 m)$

— all further elements j in ØV (IS, j) formed N groups of 3 numbers. The first number of each group is the coefficient of the product $A \cdot B$, the second and third are α and β, parameters of the appropriate integrals A (α,p) and B (β, pt), respectively.

For example the Lofthus formula for the p-pσ integral (IS = 5): s (1pσ, 1pσ) = = 3/2 ($-A_2B_2 + A_0B_0$) is stored in the program as: DATA (ØV (5, J), J = = 1,8)/1.5, 8., — 1., 2., 2., 1., 0., 0./

The subroutine ØVERLAP with a very small main program running at CDC Cyber 72 computer requires 9000 words of central memory. In the Appendix II a complete listing of the subroutine ØVERLAP is shown together with a sample of the complete printout. The complete printout is made according to the equation (2): N_{ab} is given as FACTØR, first 19 coefficients C_m as EXPANSIØN CØEFFICIENTS, and finally $A(l_i + s_a + s_b - m, p)$ and $B(k_i + m, pt)$ for m = 0 are given in MASTER FØRMULA. (IØ = 3 ... see comments at the beginning of the subroutine ØVERLAP).

Acknowledgement. The financial support of the Boris Kidrič Fund is gratefully acknowledged.

REFERENCES

1. H. H. Jaffé and G. O. Doak, *J. Chem. Phys.* **21** (1953) 196.
2. H. H. Jaffé, *J. Chem. Phys.* **21** (1953) 256.
3. J. L. Roberts and H. H. Jaffé, *J. Chem. Phys.* **27** (1957) 883.
4. L. Leifer, F. A. Cotton and J. R. Leto, *J. Chem. Phys.* **28** (1958) 364; *ibid.* **28** (1958) 1253.
5. A. Lofthus, *Mol. Phys.* **5** (1962) 105.
6. R. S. Mulliken, C. A. Rieke, D. Orloff and H. Orloff, *J. Chem. Phys.* **17** (1949) 1248.
7. For example:
   J. A. Pople and D. L. Beveridge, *Approximate Molecular Orbital Theory,* McGraw Hill, New York 1970, p. 163.
   A. C. Wahl and R. H. Land, *J. Chem. Phys.* **50** (1969) 4725.
8. J. Miller, J. M. Gerhauser and F. A. Matsen, »*Quantum Chemistry Integrals and Tables*«, University of Texas Press, Austin, Texas 1959.
9. M. Abramowitzc and I. A. Stegun, Eds., *Handbook of Mathematical Functions,* Dover Publication Inc., New York 1965, pp. 504, 260, 267.

IZVLEČEK

Računalniški program za računanje prekrivalnih integralov

J. Zupan

Opisan in diskutiran je program v jeziku FORTRAN IV za računanje σ, π, δ in φ tipov integralov prekrivanja med dvema atomskima orbitalama Slaterjevega tipa (s, p, d, f) s kvantnimi števili 1 do 6.

KEMIJSKI INŠTITUT
BORISA KIDRIČA,
61000 LJUBLJANA

Sprejeto 24. maja 1974.

APPENDIX I

The integral A defined as $A(\alpha, p) = \int_1^\infty \xi^\alpha e^{-p} \cdot d\xi$ can be written in terms of incomplete gamma function:

$$A(\alpha, p) = \frac{1}{p^{\alpha+1}} \Gamma(\alpha \pm 1, p) = \frac{1}{p^{\alpha+1}} \cdot$$

$$[\Gamma(\alpha + 1) - \gamma(\alpha + 1, p)] \quad (3)$$

where $\gamma(a, x)$ can be expressed as the confluent hypergeometric Kummer's function M[9]:

$$\gamma(a, x) = \frac{x^a e^{-x}}{a} M(1, 1 + a, x); \quad (4)$$

$$M(a, b, z) = 1 + \frac{az}{b} + \frac{(a)_2 z^2}{(b)_2 2!} + \ldots \frac{(a)_m z^n}{(b)_n n!} + \ldots$$

where

$$(a)_n = a \cdot (a + 1)(a + 2) \ldots (a + n - 1)$$
$$(a)_0 = 1$$

In our case, where $b = \alpha + 2$ and $a = 1$, Kummer's function M can be written as a sum

$$M(1, \alpha + 2, p) = 1 + \sum_{n=2}^{\infty} c_n \quad (5)$$

$$c_n = c_{n-1} \left(\frac{p}{\alpha + n}\right) \; ; \; c_1 = 1$$

Inserting equations (5) and (4) into equation (3) we get a final expression:

$$A(\alpha, p) = \frac{\alpha!}{p^{\alpha+1}} - \frac{e^{-p}}{\alpha + 1} \left[1 + \sum_{n=2}^{\infty} c_{n-1} \left(\frac{p}{\alpha + n} \right) \right]$$

where $c_1 = 1$ and $\alpha \neq -1$

in which the integral A is suitable for numerical calculation. The integrals $A(\alpha, p)$ were actually calculated only for the values of α in the interval (0,1], all the others were obtained by the recursion formulas[8]:

$$A(\alpha, p) = \frac{1}{p} [e^{-p} + \alpha A(\alpha - 1, p)]$$

$$A(\alpha, p) = \frac{1}{\alpha + 1} [p A(\alpha\ 1, p) - e^{-p}]; \; \alpha \neq -1$$

APPENDIX II

```
      SUBROUTINE OVERLAP(N1,N2,LOR1,LOR2,P,T,FACTOR,SOVE,IS,IO)
C
C*****************************************************************************
C
C     THE SUBROUTINE OVERLAP(N1,N2,LOR1,LOR2,P,T,FACTOR,SOVE,IS,IO)
C     COMPUTE ANY SIGMA, PI, DELTA OR FI TYPE OVERLAP BETWEEN TWO S, P, D
C     OR F SLATER TYPE ATOMIC ORBITALS WITH QUANTUM NUMBERS ONE TO SIX.
C
C*****************************************************************************
C
C     THE MEANING OF FIELD PARAMETERS -
C
C     N1 AND N2           QUANTUM NUMBERS FOR BOTH ATOMIC ORBITALS  (1-6)
C
C     LOR1 AND LOR2       ORBITAL QUANTUM NUMBERS -      0 FOR S
C                                                        1 FOR P
C                                                        2 FOR D
C                                                        3 FOR F
C
C     P                   MULLIKEN OVERLAP PARAMETER P=(MU1+MU2)*R/(2*R0)
C
C     T                   MULLIKEN OVERLAP PARAMETER T=(MU1-MU2)/(MU1+MU2)
C
C     IS                  NUMBER OF LOFTHUS FORMULA WHICH HAVE TO BE PERFORMED.
C                         THIS PARAMETER DEFINES THE TWO DIMENSIONAL ARRAY
C                         OV(IS,J)
C
C                         1 FOR S-S SIGMA     8 FOR D-D SIGMA      15 FOR D-F PI
C                         2 FOR S-P SIGMA     9 FOR D-F SIGMA      16 FOR F-F PI
C                         3 FOR S-D SIGMA    10 FOR F-F SIGMA      17 FOR D-D DELTA
C                         4 FOR S-F SIGMA    11 FOR P-P PI         18 FOR D-F DELTA
C                         5 FOR P-P SIGMA    12 FOR P-D PI         19 FOR F-F DELTA
C                         6 FOR P-D SIGMA    13 FOR P-F PI         20 FOR F-F FI
C                         7 FOR P-F SIGMA    14 FOR D-D PI
C
C     IO                  OUTPUT SPECIFICATION
C
C                         0 FOR NO OUTPUT, OVERLAP RETURNING VIA PARAMETER SOVE
C                         1 FOR PRINTED OVERLAP ONLY
C                         2 FOR PRINTED OVERLAP, MASTER FORMULA AND EXPANSION
C                               COEFFICIENTS
C                         3 FOR COMPLETE OUTPUT - OVERLAP, EXPANSION COEFFICIENTS,
C                               MASTER FORMULA, A AND B INTEGRALS
C                         4 FOR OVERLAP PUNCHED  ON CARDS (TAPE 5 - F10.7)
C
C     FACTOR              RETURNS THE NORMALISATION FACTOR
C
C     SOVE                RETURNS THE RESULTED OVERLAP INTEGRAL
C
C*****************************************************************************
C
      INTEGER S(500),UM(500)
      DIMENSION OV(20,56),AN(20),BN(20),CN(20),SLAT(6),FAK(6),FIN(500)
      DIMENSION TEKST(20)
      DATA FAK /2.,24.,720.,11405.8878,40320.,95809.45769/
      DATA SLAT /1.,2.,3.,3.7,4.,4.2/
      DATA TEKST/9HS-S SIGMA,9HS-P SIGMA,9HS-D SIGMA,9HS-F SIGMA,9HP-P S
     *IGMA,9HP-D SIGMA,9HP-F SIGMA,9HD-D SIGMA,9HD-F SIGMA,9HF-F SIGMA,
     *9HP-P PI   ,9HP-D PI   ,9HP-F PI   ,9HD-D PI   ,9HD-F PI   ,9HF-F
     *PI   ,9HD-D DELTA,9HD-F DELTA,9HF-F DELTA,9HF-F FI       /
      DATA(OV(1,J),J=1,5)/0.5,5.,1.,0.,0./
      DATA(OV(2,J),J=1,8)/0.86602540378,8.,-1.,1.,1.,1.,0.,0./
      DATA(OV(3,J),J=1,17)/0.55901799437,17.,-1.,2.,0.,3.,2.,2.,-4.,1.,1
     *.,3.,0.,0.,-1.,0.,2./
      DATA(OV(4,J),J=1,26)/0.66143782777,26.,3.,3.,1.,-5.,3.,3.,-3.,2.,
     *0.,9.,2.,2.,-9.,1.,1.,3.,1.,3.,5.,0.,0.,-3.,0.,2./
      DATA(OV(5,J),J=1,8)/1.5,8.,-1.,2.,2.,1.,0.,0./
```

```
      DATA(OV(6,J),J=1,26)/0.96824583655,26.,-1.,3.,1.,3.,3.,3.,-1.,2.,0
     *.,-1.,2.,2.,-1.,1.,1.,-1.,1.,3.,3.,0.,0.,-1.,0.,2./
      DATA(OV(7,J),J=1,26)/1.14564392373,26.,3.,4.,2.,-5.,4.,4.,4.,3.,
     *3.,-3.,2.,0.,3.,2.,4.,-4.,1.,1.,5.,0.,0.,-3.,0.,2./
      DATA(OV( 8,J),J=1,29)/0.625,29.,1.,4.,0.,-6.,4.,2.,9.,4.,4.,-6.,
     *2.,0.,4.,2.,2.,-6.,2.,4.,9.,0.,0.,-6.,0.,2.,1.,0.,4./
      DATA(OV(9,J),J=1,56)/0.73950997288,56.,-3.,5.,1.,14.,5.,3.,-15.,
     *5.,5.,3.,4.,0.,-6.,4.,2.,7.,4.,4.,6.,3.,1.,-12.,3.,3.,14.,3.,5.,
     *-14.,2.,0.,12.,2.,2.,-6.,2.,4.,-7.,1.,1.,6.,1.,3.,-3.,1.,5.,15.,
     *0.,0.,-14.,0.,2.,3.,0.,4./
      DATA(OV(10,J),J=1,38)/0.875,38.,-9.,6.,2.,30.,6.,4.,-25.,6.,6.,
     *9.,4.,0.,-27.,4.,4.,30.,4.,6.,-30.,2.,0.,27.,2.,2.,-9.,2.,6.,25.,
     *0.,0.,-30.,0.,2.,9.,0.,4./
      DATA(OV(11,J),J=1,14)/0.75,14.,1.,2.,0.,-1.,0.,0.,-1.,2.,2.,1.,
     *0.,2./
      DATA(OV(12,J),J=1,26)/1.67705098312,26.,-1.,3.,1.,1.,3.,3.,1.,2.,0
     *.,-1.,2.,2.,1.,1.,1.,-1.,1.,3.,-1.,0.,0.,1.,0.,2./
      DATA(OV(13,J),J=1,41)/0.70156076001,41.,-1.,4.,0.,6.,4.,2.,-5.,4.,
     *4.,-8.,3.,1.,8.,3.,3.,6.,2.,0.,-12.,2.,2.,6.,2.,4.,8.,1.,1.,-8.,1.
     *,3.,-5.,0.,0.,6.,0.,2.,-1.,0.,4./
      DATA(OV(14,J),J=1,20)/3.75,20.,-1.,4.,2.,1.,4.,4.,1.,2.,0.,-1.,2.,
     *4.,-1.,0.,0.,1.,0.,2./
      DATA(OV(15,J),J=1,56)/1.56873754975,56.,-1.,5.,1.,6.,5.,3.,-5.,
     *5.,5.,-1.,4.,0.,-2.,4.,2.,3.,4.,4.,-2.,3.,1.,-4.,3.,3.,6.,3.,5.,
     *6.,2.,0.,-4.,2.,2.,-2.,2.,4.,3.,1.,1.,-2.,1.,3.,-1.,1.,5.,-5.,0.,
     *0.,6.,0.,2.,-1.,0.,4./
      DATA(OV(16,J),J=1,50)/0.65625,50.,1.,6.,0.,-11.,6.,2.,35.,6.,4.,
     *-25.,6.,6.,-11.,4.,0.,9.,4.,2.,-33.,4.,4.,35.,4.,6.,35.,2.,0.,
     *-33.,2.,2.,9.,2.,4.,-11.,2.,6.,-25.,0.,0.,35.,0.,2.,-11.,0.,4.,
     *1.,0.,6./
      DATA(OV(17,J),J=1,29)/0.9375,29.,1.,4.,0.,-2.,2.,0.,1.,0.,0.,-2.,
     *4.,2.,4.,2.,2.,-2.,0.,2.,1.,4.,4.,-2.,2.,4.,1.,0.,4./
      DATA(OV(18,J),J=1,56)/2.48039185412,56.,1.,4.,0.,-2.,2.,0.,1.,0.,
     *0.,-2.,4.,2.,4.,2.,2.,-2.,0.,2.,1.,4.,4.,-2.,2.,4.,1.,0.,4.,-1.,
     *5.,1.,2.,5.,3.,-1.,5.,5.,2.,3.,1.,-4.,3.,3.,2.,3.,5.,-1.,1.,1.,2.,
     *1.,3.,-1.,1.,5./
      DATA(OV(19,J),J=1,56)/6.5625,56.,1.,4.,0.,-2.,2.,0.,1.,0.,0.,-2.,
     *4.,2.,4.,2.,2.,-2.,0.,2.,1.,4.,4.,-2.,2.,4.,1.,0.,4.,-1.,6.,2.,
     *2.,4.,2.,-1.,2.,2.,2.,6.,4.,-4.,4.,4.,2.,2.,4.,-1.,6.,6.,2.,4.,
     *6.,-1.,2.,6./
      DATA(OV(20,J),J=1,50)/1.09375,50.,1.,6.,0.,-3.,4.,0.,3.,2.,0.,-1.,
     *0.,0.,-3.,6.,2.,9.,4.,2.,-9.,2.,2.,3.,0.,2.,3.,6.,4.,-9.,4.,4.,
     *9.,2.,4.,-3.,0.,4.,-1.,6.,6.,3.,4.,6.,-3.,2.,6.,1.,0.,6./
      IF(IO.GT.4.OR.IO.LT.0)  IO= 2
      IF(IO.NE.0)  PRINT 101
C
C     NORMALIZATION FACTOR FOR THE FINAL OVERLAP
C
      AO=P*(1+T)
      BO=P*(1-T)
      PT=P*T
      FACTOR=OV(IS,1)*(AO**(SLAT(N1)+0.5))*(BO**(SLAT(N2)+0.5))/
     *((FAK(N2)*FAK(N1))**0.5)
      NOST=OV(IS,2)
      DO 11 J=1,NOST
   11 S(J)=OV(IS,J)
C
C     DERIVATION OF THE LOFTHUS POLYNOMIALS S ON PARAMETER ALFA
C
      NDER1=SLAT(N1)-LOR1
      IF(NDER1.LE.0)  GO TO 12
      CALL DER(S,UM,NOST,1,NDER1)
   12 CONTINUE
C
C     DERIVATION OF THE LOFTHUS POLYNOMIALS S ON PARAMETER BETA
C
      NDER2=SLAT(N2)-LOR2
      IF(NDER2.LE.0)  GO TO 13
      CALL DER(S,UM,NOST,2,NDER2)
   13 CONTINUE
```

```
C
C     COMPUTATION OF EXPANSION COEFFICIENTS FOR THE CASE OF NONINTEGER
C     QUANTUM NUMBERS
C
      AN(1)=1.
      BN(1)=1.
      REL1=SLAT(N1)-IFIX(SLAT(N1))
      REL2=SLAT(N2)-IFIX(SLAT(N2))
      DO 20 J=2,20
      AN(J)=AN(J-1)*(REL1-J+2)/(J-1)
      BN(J)=-1.*BN(J-1)*(REL2-J+2)/(J-1)
      CN(J-1)=0.
      DO 20 II=2,J
      JJ=J-II+1
      CN(J-1)=CN(J-1)+AN(II-1)*BN(JJ)
   20 CONTINUE
C
C     COMPUTATION OF FINAL LOFTHUS OVERLAP POLYNOMIAL S EVALUATING  INTEGRALS
C     A(F1,P) AND B(I2,PT)  WITH PROPER ARGUMENTS
C
      SOV=0.
      NIK=19
      IF((N1.LE.3.AND.N2.LE.3).OR.(N1.LE.3.AND.N2.EQ.5).OR.
     *(N1.EQ.5.AND.N2.LE.3).OR.(N1.EQ.5.AND.N2.EQ.5)) NIK=1
      DO 21 I=1,NIK
      DO 21 J=3,NOST,3
      F1=S(J+1)-I+1+REL1+REL2
      I2=S(J+2)+I-1
      AS=A(F1,P)
      BS=B(I2,PT)
      IF(IO.EQ.3)  PRINT 100,F1,P,AS,I2,PT,BS
      SOV=SOV+AS*BS*S(J)*CN(I)
   21 CONTINUE
      SOVE=SOV*FACTOR
      IF(IO.EQ.0)  RETURN
      PRINT 108,N1,N2,TEKST(IS),P,T,FACTOR,SOV,SOVE
      IF(IO.EQ.4)  WRITE(5,105)  SOVE
      IF(IO.EQ.1)  RETURN
      PRINT 109,(CN(J),J=1,19)
      DO 22 J=3,NOST,3
      FIN(J)=S(J)
      FIN(J+1)=S(J+1)+REL1+REL2
   22 FIN(J+2)=S(J+2)
      PRINT 106,(FIN(J),J=3,NOST)
      PRINT 107
  100 FORMAT(2X,*A(*,2F5.1,*)=*,E15.7,9X,*B(*,I3,F7.2,*)=*,E15.7)
  101 FORMAT(1H1)
  105 FORMAT(F10.7)
  106 FORMAT(/,15H MASTER FORMULA,//,6(2H (,F4.0,4H)*A(,F3.1,3H)B(,F2.0
     *,1H),2H +))
  107 FORMAT(1H0,125H IN ORDER TO GET MASTER FORMULAS FOR OTHER ELEMENTS
     * IN THE SERIES EXPANSION IT IS NECESSARY TO DECREASE EACH TIME IN
     *THE         ,/,130H PRESENTED FORMULA THE ARGUMENTS F1 IN ALL INTEG
     *RALS A(F1,P) FOR 1, AND INCREASE ARGUMENTS I2 IN THE INTEGRALS B(I
     *2,PT) FOR 1.          )
  108 FORMAT(//,* S(*,I1,1H-,I1,1H,A9,1H,F3.1,1H,,F3.1,14H) = FACTOR*S =
     *,F12.6,2H *,F12.6,2H =,F9.6)
  109 FORMAT(/,* EXPANSION COEFFICIENTS *,/,10(1X,F10.6))
      RETURN
      END
```

```
      SUBROUTINE DER(S,UM,NOST,N1,N2)
C
C     DERIVATION OF LOFTHUS POLYNOMIALS S ACCORDING TO THE RULES DESCRIBED IN
C     A. LOFTHUS, MOL. PHYS., 5, (1962), P. 105-114
C
      INTEGER S(1),UM(1)
      DO 1 J=1,N2
      DO 2 JJ=3,NOST,3
      J2=2*(JJ-1)-1
      UM(J2)=S(JJ)
      UM(J2+1)=S(JJ+1)+1
      UM(J2+2)=S(JJ+2)
      UM(J2+4)=S(JJ+1)
      UM(J2+5)=S(JJ+2)+1
      UM(J2+3)=S(JJ)
      IF(N1.EQ.2) UM(J2+3)=-UM(J2+3)
    2 CONTINUE
      NOST=2*(NOST-2)+2
      CALL DUMP(UM,NOST)
      DO 6 JJ=3,NOST
    6 S(JJ)=UM(JJ)
    1 CONTINUE
      RETURN
      END


      SUBROUTINE DUMP(UM,NOST)
C
C     COMPRESSION OF THE POLYNOMIAL OBTAINED BY DERIVATION IN SUBROUTINE DER
C
      INTEGER UM(1)
      DIMENSION  IP(500)
      DO 20  K=3,NOST,3
      KK=K+3
      IF(UM(K).EQ.0) GO TO 20
      DO 21 L=KK,NOST,3
      IF(UM(K+1).EQ.UM(L+1).AND.UM(K+2).EQ.UM(L+2)) GO TO 22
      GO TO 21
   22 UM(K)=UM(K)+UM(L)
      UM(L)=0.
   21 CONTINUE
   20 CONTINUE
      IT=2
      DO 23 K=3,NOST,3
      IF(UM(K).EQ.0) GO TO 23
      IT=IT+1
      IP(IT)=UM(K)
      IT=IT+1
      IP(IT)=UM(K+1)
      IT=IT+1
      IP(IT)=UM(K+2)
   23 CONTINUE
      DO 25 K=3,IT
   25 UM(K)=IP(K)
      NOST=IT
      RETURN
      END
```

```
A(  5.7  2.0)=  .3949327E+01      B(  0   1.00)=   .2350402E+01
A(  4.7  2.0)=  .1361986E+01      B(  1   1.00)=  -.7357589E+00
A(  3.7  2.0)=  .5507737E+00      B(  2   1.00)=   .8788846E+00
A(  2.7  2.0)=  .2611384E+00      B(  3   1.00)=  -.4495074E+00
A(  1.7  2.0)=  .1433117E+00      B(  4   1.00)=   .5523728E+00
A(   .7  2.0)=  .8899300E-01      B(  5   1.00)=  -.3242974E+00
A(  4.7  2.0)=  .1361986E+01      B(  1   1.00)=  -.7357589E+00
A(  3.7  2.0)=  .5507737E+00      B(  2   1.00)=   .8788846E+00
A(  2.7  2.0)=  .2611384E+00      B(  3   1.00)=  -.4495074E+00
A(  1.7  2.0)=  .1433117E+00      B(  4   1.00)=   .5523728E+00
A(   .7  2.0)=  .8899300E-01      B(  5   1.00)=  -.3242974E+00
A(  -.3  2.0)=  .6092960E-01      B(  6   1.00)=   .4046182E+00
A(  3.7  2.0)=  .5507737E+00      B(  2   1.00)=   .8788846E+00
A(  2.7  2.0)=  .2611384E+00      B(  3   1.00)=  -.4495074E+00
A(  1.7  2.0)=  .1433117E+00      B(  4   1.00)=   .5523728E+00
A(   .7  2.0)=  .8899300E-01      B(  5   1.00)=  -.3242974E+00
A(  -.3  2.0)=  .6092960E-01      B(  6   1.00)=   .4046182E+00
A( -1.3  2.0)=  .4492030E-01      B(  7   1.00)=  -.2538341E+00
A(  2.7  2.0)=  .2611384E+00      B(  3   1.00)=  -.4495074E+00
A(  1.7  2.0)=  .1433117E+00      B(  4   1.00)=   .5523728E+00
A(   .7  2.0)=  .8899300E-01      B(  5   1.00)=  -.3242974E+00
A(  -.3  2.0)=  .6092960E-01      B(  6   1.00)=   .4046182E+00
A( -1.3  2.0)=  .4492030E-01      B(  7   1.00)=  -.2538341E+00
A( -2.3  2.0)=  .3499591E-01      B(  8   1.00)=   .3197297E+00
A(  1.7  2.0)=  .1433117E+00      B(  4   1.00)=   .5523728E+00
A(   .7  2.0)=  .8899300E-01      B(  5   1.00)=  -.3242974E+00
A(  -.3  2.0)=  .6092960E-01      B(  6   1.00)=   .4046182E+00
A( -1.3  2.0)=  .4492030E-01      B(  7   1.00)=  -.2538341E+00
A( -2.3  2.0)=  .3499591E-01      B(  8   1.00)=   .3197297E+00
A( -3.3  2.0)=  .2841020E-01      B(  9   1.00)=  -.2085940E+00
A(   .7  2.0)=  .8899300E-01      B(  5   1.00)=  -.3242974E+00
A(  -.3  2.0)=  .6092960E-01      B(  6   1.00)=   .4046182E+00
A( -1.3  2.0)=  .4492030E-01      B(  7   1.00)=  -.2538341E+00
A( -2.3  2.0)=  .3499591E-01      B(  8   1.00)=   .3197297E+00
A( -3.3  2.0)=  .2841020E-01      B(  9   1.00)=  -.2085940E+00
A( -4.3  2.0)=  .2379239E-01      B( 10   1.00)=   .2644629E+00
A(  -.3  2.0)=  .6092960E-01      B(  6   1.00)=   .4046182E+00
A( -1.3  2.0)=  .4492030E-01      B(  7   1.00)=  -.2538341E+00
A( -2.3  2.0)=  .3499591E-01      B(  8   1.00)=   .3197297E+00
A( -3.3  2.0)=  .2841020E-01      B(  9   1.00)=  -.2085940E+00
A( -4.3  2.0)=  .2379239E-01      B( 10   1.00)=   .2644629E+00
A( -5.3  2.0)=  .2040709E-01      B( 11   1.00)=  -.1770699E+00
A( -1.3  2.0)=  .4492030E-01      B(  7   1.00)=  -.2538341E+00
A( -2.3  2.0)=  .3499591E-01      B(  8   1.00)=   .3197297E+00
A( -3.3  2.0)=  .2841020E-01      B(  9   1.00)=  -.2085940E+00
A( -4.3  2.0)=  .2379239E-01      B( 10   1.00)=   .2644629E+00
A( -5.3  2.0)=  .2040709E-01      B( 11   1.00)=  -.1770699E+00
A( -6.3  2.0)=  .1783417E-01      B( 12   1.00)=   .2255634E+00
A( -2.3  2.0)=  .3499591E-01      B(  8   1.00)=   .3197297E+00
A( -3.3  2.0)=  .2841020E-01      B(  9   1.00)=  -.2085940E+00
A( -4.3  2.0)=  .2379239E-01      B( 10   1.00)=   .2644629E+00
A( -5.3  2.0)=  .2040709E-01      B( 11   1.00)=  -.1770699E+00
A( -6.3  2.0)=  .1783417E-01      B( 12   1.00)=   .2255634E+00
A( -7.3  2.0)=  .1582015E-01      B( 13   1.00)=  -.1538375E+00
A( -3.3  2.0)=  .2841020E-01      B(  9   1.00)=  -.2085940E+00
A( -4.3  2.0)=  .2379239E-01      B( 10   1.00)=   .2644629E+00
A( -5.3  2.0)=  .2040709E-01      B( 11   1.00)=  -.1770699E+00
A( -6.3  2.0)=  .1783417E-01      B( 12   1.00)=   .2255634E+00
A( -7.3  2.0)=  .1582015E-01      B( 13   1.00)=  -.1538375E+00
A( -8.3  2.0)=  .1420479E-01      B( 14   1.00)=   .1966773E+00
A( -4.3  2.0)=  .2379239E-01      B( 10   1.00)=   .2644629E+00
A( -5.3  2.0)=  .2040709E-01      B( 11   1.00)=  -.1770699E+00
A( -6.3  2.0)=  .1783417E-01      B( 12   1.00)=   .2255634E+00
A( -7.3  2.0)=  .1582015E-01      B( 13   1.00)=  -.1538375E+00
A( -8.3  2.0)=  .1420479E-01      B( 14   1.00)=   .1966773E+00
A( -9.3  2.0)=  .1288261E-01      B( 15   1.00)=  -.1360020E+00
A( -5.3  2.0)=  .2040709E-01      B( 11   1.00)=  -.1770699E+00
A( -6.3  2.0)=  .1783417E-01      B( 12   1.00)=   .2255634E+00
A( -7.3  2.0)=  .1582015E-01      B( 13   1.00)=  -.1538375E+00
A( -8.3  2.0)=  .1420479E-01      B( 14   1.00)=   .1966773E+00
A( -9.3  2.0)=  .1288261E-01      B( 15   1.00)=  -.1360020E+00
A(-10.3  2.0)=  .1178173E-01      B( 16   1.00)=   .1743697E+00
A( -6.3  2.0)=  .1783417E-01      B( 12   1.00)=   .2255634E+00
A( -7.3  2.0)=  .1582015E-01      B( 13   1.00)=  -.1538375E+00
A( -8.3  2.0)=  .1420479E-01      B( 14   1.00)=   .1966773E+00
A( -9.3  2.0)=  .1288261E-01      B( 15   1.00)=  -.1360020E+00
A(-10.3  2.0)=  .1178173E-01      B( 16   1.00)=   .1743697E+00
A(-11.3  2.0)=  .1085163E-01      B( 17   1.00)=  -.1218769E+00
A( -7.3  2.0)=  .1582015E-01      B( 13   1.00)=  -.1538375E+00
A( -8.3  2.0)=  .1420479E-01      B( 14   1.00)=   .1966773E+00
A( -9.3  2.0)=  .1288261E-01      B( 15   1.00)=  -.1360020E+00
A(-10.3  2.0)=  .1178173E-01      B( 16   1.00)=   .1743697E+00
A(-11.3  2.0)=  .1085163E-01      B( 17   1.00)=  -.1218769E+00
A(-12.3  2.0)=  .1005593E-01      B( 18   1.00)=   .1566184E+00
A( -8.3  2.0)=  .1420479E-01      B( 14   1.00)=   .1966773E+00
A( -9.3  2.0)=  .1288261E-01      B( 15   1.00)=  -.1360020E+00
A(-10.3  2.0)=  .1178173E-01      B( 16   1.00)=   .1743697E+00
A(-11.3  2.0)=  .1085163E-01      B( 17   1.00)=  -.1218769E+00
A(-12.3  2.0)=  .1005593E-01      B( 18   1.00)=   .1566184E+00
A(-13.3  2.0)=  .9367758E-02      B( 19   1.00)=  -.1104124E+00
A( -9.3  2.0)=  .1288261E-01      B( 15   1.00)=  -.1360020E+00
A(-10.3  2.0)=  .1178173E-01      B( 16   1.00)=   .1743697E+00
A(-11.3  2.0)=  .1085163E-01      B( 17   1.00)=  -.1218769E+00
A(-12.3  2.0)=  .1005593E-01      B( 18   1.00)=   .1566184E+00
A(-13.3  2.0)=  .9367758E-02      B( 19   1.00)=  -.1104124E+00
A(-14.3  2.0)=  .8766900E-02      B( 20   1.00)=   .1421543E+00
A(-10.3  2.0)=  .1178173E-01      B( 16   1.00)=   .1743697E+00
A(-11.3  2.0)=  .1085163E-01      B( 17   1.00)=  -.1218769E+00
A(-12.3  2.0)=  .1005593E-01      B( 18   1.00)=   .1566184E+00
A(-13.3  2.0)=  .9367758E-02      B( 19   1.00)=  -.1104124E+00
A(-14.3  2.0)=  .8766900E-02      B( 20   1.00)=   .1421543E+00
A(-15.3  2.0)=  .8237866E-02      B( 21   1.00)=  -.1009210E+00
A(-11.3  2.0)=  .1085163E-01      B( 17   1.00)=  -.1218769E+00
A(-12.3  2.0)=  .1005593E-01      B( 18   1.00)=   .1566184E+00
A(-13.3  2.0)=  .9367758E-02      B( 19   1.00)=  -.1104124E+00
A(-14.3  2.0)=  .8766900E-02      B( 20   1.00)=   .1421543E+00
A(-15.3  2.0)=  .8237866E-02      B( 21   1.00)=  -.1009210E+00
A(-16.3  2.0)=  .7768598E-02      B( 22   1.00)=   .1301403E+00
A(-12.3  2.0)=  .1005593E-01      B( 18   1.00)=   .1566184E+00
A(-13.3  2.0)=  .9367758E-02      B( 19   1.00)=  -.1104124E+00
A(-14.3  2.0)=  .8766900E-02      B( 20   1.00)=   .1421543E+00
A(-15.3  2.0)=  .8237866E-02      B( 21   1.00)=  -.1009210E+00
A(-16.3  2.0)=  .7768598E-02      B( 22   1.00)=   .1301403E+00
A(-17.3  2.0)=  .7349576E-02      B( 23   1.00)=  -.9293336E-01

S(2-4,S-S SIGMA,2.0, .5) = FACTOR*S =    .014897 *    9.952035 =  .148257

EXPANSION COEFFICIENTS
  1.000000   -.700000   -.105000   -.045500   -.026162   -.017267   -.012375   -.009370   -.007379   -.005985
  -.004967   -.004200   -.003605   -.003133   -.002753   -.002441   -.002182   -.001963   -.001778

MASTER FORMULA

(  1.)*A(5.7)B(0.) + ( -1.)*A(4.7)B(1.) + ( -2.)*A(3.7)B(2.) + (  2.)*A(2.7)B(3.) + (  1.)*A(1.7)B(4.) + ( -1.)*A( .7)B(5.) +

 IN ORDER TO GET MASTER FORMULAS FOR OTHER ELEMENTS IN THE SERIES EXPANSION IT IS NECESSARY TO DECREASE FACH TIME IN THE
PRESENTED FORMULA THE ARGUMENTS F1 IN ALL INTEGRALS A(F1,P) FOR 1, AND INCREASE ARGUMENTS I2 IN THE INTEGRALS B(I2,PT) FOR 1.
```